
1 [bookmark: _Toc122434485][bookmark: _Toc144288531]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of NR5GC Redcap test case 7.1.1.1.17 in FR1 which is part of the NR5GC test suite in iWD_TTCN3-B2022-09_D23wk24.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	Parikshit.bhise@rohde-schwarz.com 
	
1. [bookmark: _Toc144288532]Verification Test Summary 
Test Case: 	7.1.1.1.17.NR5GC
ATS Version:	iwd-TTCN3-B2022-09_D23wk24
[bookmark: _Hlk37919671]System Simulator used:	R&S® 5G Protocol Conformance Test platform
UE used:	MediaTek MT 6815
Verification Status:	PASS

2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc144288533]Corrections required
2.1 [bookmark: _Toc144288534]function f_NR_DciDlInfo_Init
	Function name
	f_NR_DciDlInfo_Init

	Reason for change
	1. SIB1 size considering initialUplinkBWP-RedCap and initialDownlinkBWP-RedCap has gone beyond 1032 bit thus existing 7 RBs allocation for SIB1 is not sufficient to cater the required TBS size for its transmission. 

	Summary of change
	1. 9 RBs allocation is required to cater the TBS size requirement for these new Rel 17 test cases. Thus, made the change to use 9 RBs allocation instead of 7 for such special test cases by adding a new method called f_NR_RedcapWithIncreasedSib1Size().

Note: This will require correction in TS 38.523-3 tables as well.

	TTCN module
	Common\NR\NR_DCI_Configuration.ttcn

	MCC160 Comment
	



Before Change:
	  function f_NR_DciDlInfo_Init(NR_PhysicalParameters_Type p_PhysicalParameters,
                               template (value) NR_DciFormat_1_X_SpecificInfo_Type p_DciFormatAndParams,
                               template (value) NR_RedundancyVersionList_Type p_RedundancyVersionList,
                               boolean p_DciFormat1_0_Ccch := false)
    return template (value) NR_DciDlInfo_Type
    { /* NOTE: In general the given p_PhysicalParameters are preconfigured in TTCN and the DciInfo is pre-configured at the SS before an RRC connection gets established at the UE */
      // @sic R5-202677 : added p_DciFormat1_0_Ccch sic@
    var template (value) NR_FreqDomainResourceAssignmentDL_Type v_FreqDomainResourceAssignmentDL;
    var template (omit) PDSCH_TimeDomainResourceAllocationList v_TimeDomainResourceAllocationList := omit;
    var NR_DownlinkBWP_Type v_DownlinkBWP := p_PhysicalParameters.DL_BWPs[0];
    var PDSCH_Config v_PDSCH_Config := v_DownlinkBWP.Pdsch.ConfigDedicated.R15;
    var PDSCH_ConfigCommon v_PDSCH_ConfigCommon := v_DownlinkBWP.Pdsch.ConfigCommon.R15;
    var NR_ResourceAllocationType_Type v_ResourceAllocationType := resourceAllocationType1;
    var template (value) NR_HarqProcessConfig_Type v_HarqProcessConfig;
    var NR_ModulationSchemePDSCH_Type v_ModulationScheme;
    var integer v_MaxRbCnt := 24; // minimum value of Maximum transmission bandwidth (taking into account CORESET#0)
    var integer v_TimeDomainResourceAssignmentTableSize := 16;
    var integer v_TimeDomainResourceAssignmentIndexValue := 0;    /* => first entry of the TimeDomainResourceAssignmentTable:
                                                                     NOTE 1: if there is no pdsch_TimeDomainAllocationList configured at the UE, 38.508-1 Table 5.1.2.1.1-2 applies
                                                                     NOTE 2: according to 38.508-1 Table 5.1.2.1.1-1 for SIB1 always 8.508-1 Table 5.1.2.1.1-2 applies */
    var template (value) NR_DciCommon_TimeDomainResourceAssignment_Type v_TimeDomainResourceAssignment;
    var boolean v_IsDciFormat_1_1 := ischosen(p_DciFormatAndParams.Format_1_1);
    var integer v_FirstRbIndex;
    var integer v_FirstRbIndexSI;
    var SSB_Index v_SSB_Index := f_NR_ConvertPositionsInBurst_ToSSB_Index(p_PhysicalParameters.SSB_BurstConfig.PositionsInBurst);  //@sic R5-201150 sic@
    var integer v_FirstRbIndexDcchDtch;  //@sic R5-202677  sic@
    
    // Set RB start as per 38.523-3 Table 7.1.2.2.3.1-1: Resource allocation for frequency domain multiplexing, RA type1, non-interleaved @sic R5-201150 sic@
    select (v_SSB_Index) {
        case(tsc_NR_SSB_Index1) {
            v_FirstRbIndex := 7;
            v_FirstRbIndexSI := 0;
            v_FirstRbIndexDcchDtch := 7 + p_PhysicalParameters.OffsetCarrierCORESET0;
        }
        case(tsc_NR_SSB_Index0) {
            v_FirstRbIndex := 0;
            v_FirstRbIndexSI := v_MaxRbCnt-7;
            v_FirstRbIndexDcchDtch := 0;
        }
    }
    if (v_IsDciFormat_1_1) {
      // resource allocation type for DCI format 1_1:
      if (match(resourceAllocationType0, v_PDSCH_Config.resourceAllocation)) {  // 38.214 clause 6.1.2.2: allocation type 0 is applicable for DCI format 0_1 only
        v_ResourceAllocationType := resourceAllocationType0;
      }
      
      // size of pdsch_TimeDomainAllocationList for DCI format 1_1:
      if (ischosen(v_PDSCH_Config.pdsch_TimeDomainAllocationList.setup)) {      // 38.214 Table 5.1.2.1.1-1: pdsch-TimeDomainAllocationList provided in pdsch-Config
        v_TimeDomainResourceAllocationList := v_PDSCH_Config.pdsch_TimeDomainAllocationList.setup;
      } else {                                                                  // 38.214 Table 5.1.2.1.1-1: pdsch-TimeDomainAllocationList provided in pdsch-ConfigCommon
        v_TimeDomainResourceAllocationList := v_PDSCH_ConfigCommon.pdsch_TimeDomainAllocationList;      /* NOTE: pdsch-TimeDomainAllocationList may be not present in PUSCH-ConfigCommon */
      }
      if (isvalue(v_TimeDomainResourceAllocationList)) {
        v_TimeDomainResourceAssignmentTableSize := lengthof(v_TimeDomainResourceAllocationList);
      } else {
        v_TimeDomainResourceAssignmentTableSize := 16;        /* Default PDSCH time domain resource allocation (38.214 Table 5.1.2.1.1-2/3/4/5) */
      }
    }
    v_TimeDomainResourceAssignment := fl_NR_DciCommon_TimeDomainResourceAssignment(v_TimeDomainResourceAssignmentTableSize, v_TimeDomainResourceAssignmentIndexValue);

    select union (valueof(p_DciFormatAndParams)) {
      case (Format_1_0, Format_1_1) {   /* UE-specific search space (C-RNTI, CS-RNTI(s)) or
                                           Type1-PDCCH common search space for scheduling of Msg4 of RACH procedure (TC-RNTI) */
        if (p_DciFormat1_0_Ccch) {
            v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);
        } else {
            v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexDcchDtch, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1; may be changed dynamically during a test case @sic R5-192813 sic@ */
        }
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_AnyProcess;                                                      /* may be changed dynamically during a test case */
        v_ModulationScheme := qam64;                                                                                    /* up to 64QAM only per default */
        if (v_IsDciFormat_1_1 and match(qam256, v_PDSCH_Config.mcs_Table)) {
          v_ModulationScheme := qam256;                                                                                 /* mcs-Table given by PDSCH-Config is set to '256QAM' AND DCI format 1_1
                                                                                                                           => MCS index table 2 with modulation scheme up to 256QAM */
        }
      }
      case (Format_1_0_P_RNTI) {        /* Type2-PDCCH common search space for scheduling of Paging messages (P-RNTI) */
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1 @sic R5-192813 sic@ */
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_None;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case (Format_1_0_SI_RNTI) {       /* Type0-PDCCH common search space for scheduling of SIB1 (SI-RNTI) or
                                           Type0A-PDCCH common search space for scheduling of other system information (SI-RNTI) */
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexSI, 7);                       /* according to 38.523-3 Table 7.1.2.2.3.1-1 and Table 7.3.3.2-1 @sic R5-192812, R5-192813 sic@ */
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_Broadcast;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case (Format_1_0_RA_RNTI) {
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1 @sic R5-192813 sic@ */
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_None;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case else {
        FatalError(__FILE__, __LINE__, "unknown or invalid DCI format");
      }
    }
    return cs_NR_DciDlInfo_Init(v_ResourceAllocationType,
                                v_FreqDomainResourceAssignmentDL,
                                v_TimeDomainResourceAssignment,
                                cs_NR_TransportBlockSchedulingDL_Automatic(v_ModulationScheme, p_RedundancyVersionList),
                                v_HarqProcessConfig,
                                p_DciFormatAndParams);
  }



After Change:
	function f_NR_DciDlInfo_Init(NR_PhysicalParameters_Type p_PhysicalParameters,
                               template (value) NR_DciFormat_1_X_SpecificInfo_Type p_DciFormatAndParams,
                               template (value) NR_RedundancyVersionList_Type p_RedundancyVersionList,
                               boolean p_DciFormat1_0_Ccch := false)
    return template (value) NR_DciDlInfo_Type
    { /* NOTE: In general the given p_PhysicalParameters are preconfigured in TTCN and the DciInfo is pre-configured at the SS before an RRC connection gets established at the UE */
      // @sic R5-202677 : added p_DciFormat1_0_Ccch sic@
    var template (value) NR_FreqDomainResourceAssignmentDL_Type v_FreqDomainResourceAssignmentDL;
    var template (omit) PDSCH_TimeDomainResourceAllocationList v_TimeDomainResourceAllocationList := omit;
    var NR_DownlinkBWP_Type v_DownlinkBWP := p_PhysicalParameters.DL_BWPs[0];
    var PDSCH_Config v_PDSCH_Config := v_DownlinkBWP.Pdsch.ConfigDedicated.R15;
    var PDSCH_ConfigCommon v_PDSCH_ConfigCommon := v_DownlinkBWP.Pdsch.ConfigCommon.R15;
    var NR_ResourceAllocationType_Type v_ResourceAllocationType := resourceAllocationType1;
    var template (value) NR_HarqProcessConfig_Type v_HarqProcessConfig;
    var NR_ModulationSchemePDSCH_Type v_ModulationScheme;
    var integer v_MaxRbCnt := 24; // minimum value of Maximum transmission bandwidth (taking into account CORESET#0)
    var integer v_TimeDomainResourceAssignmentTableSize := 16;
    var integer v_TimeDomainResourceAssignmentIndexValue := 0;    /* => first entry of the TimeDomainResourceAssignmentTable:
                                                                     NOTE 1: if there is no pdsch_TimeDomainAllocationList configured at the UE, 38.508-1 Table 5.1.2.1.1-2 applies
                                                                     NOTE 2: according to 38.508-1 Table 5.1.2.1.1-1 for SIB1 always 8.508-1 Table 5.1.2.1.1-2 applies */
    var template (value) NR_DciCommon_TimeDomainResourceAssignment_Type v_TimeDomainResourceAssignment;
    var boolean v_IsDciFormat_1_1 := ischosen(p_DciFormatAndParams.Format_1_1);
    var integer v_FirstRbIndex;
    var integer v_FirstRbIndexSI;
    var SSB_Index v_SSB_Index := f_NR_ConvertPositionsInBurst_ToSSB_Index(p_PhysicalParameters.SSB_BurstConfig.PositionsInBurst);  //@sic R5-201150 sic@
    var integer v_FirstRbIndexDcchDtch;  //@sic R5-202677  sic@
    var boolean v_UseIncreasedRBSib1 := false;
        
    if(f_NR_RedcapWithIncreasedSib1Size())  //WA#WI=1098062
       {
           v_UseIncreasedRBSib1 := true;
       }
        
    // Set RB start as per 38.523-3 Table 7.1.2.2.3.1-1: Resource allocation for frequency domain multiplexing, RA type1, non-interleaved @sic R5-201150 sic@
    select (v_SSB_Index) {
        case(tsc_NR_SSB_Index1) {
            
            if(v_UseIncreasedRBSib1){
            v_FirstRbIndex := 9;
            v_FirstRbIndexSI := 0;
            v_FirstRbIndexDcchDtch := 9 + p_PhysicalParameters.OffsetCarrierCORESET0;  
            }
            else {
            v_FirstRbIndex := 7;
            v_FirstRbIndexSI := 0;
            v_FirstRbIndexDcchDtch := 7 + p_PhysicalParameters.OffsetCarrierCORESET0; }
        }
        case(tsc_NR_SSB_Index0) {
            if(v_UseIncreasedRBSib1){
               v_FirstRbIndex := 0;
            v_FirstRbIndexSI := v_MaxRbCnt-9;
            v_FirstRbIndexDcchDtch := 0; 
            }
            else {
            v_FirstRbIndex := 0;
            v_FirstRbIndexSI := v_MaxRbCnt-7;
            v_FirstRbIndexDcchDtch := 0;
            }
        }
    }
    if (v_IsDciFormat_1_1) {
      // resource allocation type for DCI format 1_1:
      if (match(resourceAllocationType0, v_PDSCH_Config.resourceAllocation)) {  // 38.214 clause 6.1.2.2: allocation type 0 is applicable for DCI format 0_1 only
        v_ResourceAllocationType := resourceAllocationType0;
      }
      
      // size of pdsch_TimeDomainAllocationList for DCI format 1_1:
      if (ischosen(v_PDSCH_Config.pdsch_TimeDomainAllocationList.setup)) {      // 38.214 Table 5.1.2.1.1-1: pdsch-TimeDomainAllocationList provided in pdsch-Config
        v_TimeDomainResourceAllocationList := v_PDSCH_Config.pdsch_TimeDomainAllocationList.setup;
      } else {                                                                  // 38.214 Table 5.1.2.1.1-1: pdsch-TimeDomainAllocationList provided in pdsch-ConfigCommon
        v_TimeDomainResourceAllocationList := v_PDSCH_ConfigCommon.pdsch_TimeDomainAllocationList;      /* NOTE: pdsch-TimeDomainAllocationList may be not present in PUSCH-ConfigCommon */
      }
      if (isvalue(v_TimeDomainResourceAllocationList)) {
        v_TimeDomainResourceAssignmentTableSize := lengthof(v_TimeDomainResourceAllocationList);
      } else {
        v_TimeDomainResourceAssignmentTableSize := 16;        /* Default PDSCH time domain resource allocation (38.214 Table 5.1.2.1.1-2/3/4/5) */
      }
    }
    v_TimeDomainResourceAssignment := fl_NR_DciCommon_TimeDomainResourceAssignment(v_TimeDomainResourceAssignmentTableSize, v_TimeDomainResourceAssignmentIndexValue);

    select union (valueof(p_DciFormatAndParams)) {
      case (Format_1_0, Format_1_1) {   /* UE-specific search space (C-RNTI, CS-RNTI(s)) or
                                           Type1-PDCCH common search space for scheduling of Msg4 of RACH procedure (TC-RNTI) */
        if (p_DciFormat1_0_Ccch) {
             if(v_UseIncreasedRBSib1){
                 v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-9);
             }
            else{
            v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);
            }
        } else {
            if(v_UseIncreasedRBSib1){
            v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexDcchDtch, v_MaxRbCnt-9); 
            
            }
            else {
            v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexDcchDtch, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1; may be changed dynamically during a test case @sic R5-192813 sic@ */
               }
            }
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_AnyProcess;                                                      /* may be changed dynamically during a test case */
        v_ModulationScheme := qam64;                                                                                    /* up to 64QAM only per default */
        if (v_IsDciFormat_1_1 and match(qam256, v_PDSCH_Config.mcs_Table)) {
          v_ModulationScheme := qam256;                                                                                 /* mcs-Table given by PDSCH-Config is set to '256QAM' AND DCI format 1_1
                                                                                                                           => MCS index table 2 with modulation scheme up to 256QAM */
        }
      }
      case (Format_1_0_P_RNTI) {        /* Type2-PDCCH common search space for scheduling of Paging messages (P-RNTI) */
        if(v_UseIncreasedRBSib1){
             v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-9); 
        }
         else {
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1 @sic R5-192813 sic@ */
         }
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_None;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case (Format_1_0_SI_RNTI) {       /* Type0-PDCCH common search space for scheduling of SIB1 (SI-RNTI) or
                                           Type0A-PDCCH common search space for scheduling of other system information (SI-RNTI) */
        
         if(v_UseIncreasedRBSib1){
             v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexSI, 9); 
         }
          else {
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndexSI, 7);   /* according to 38.523-3 Table 7.1.2.2.3.1-1 and Table 7.3.3.2-1 @sic R5-192812, R5-192813 sic@ */
      }
          v_HarqProcessConfig := cs_NR_HarqProcessConfig_Broadcast;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case (Format_1_0_RA_RNTI) {
         if(v_UseIncreasedRBSib1){
             v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-9); 
         }
          else{
        v_FreqDomainResourceAssignmentDL := cs_NR_FreqDomainResourceAssignmentDL_Automatic(v_FirstRbIndex, v_MaxRbCnt-7);            /* according to 38.523-3 Table 7.1.2.2.3.1-1 @sic R5-192813 sic@ */
          }
        v_HarqProcessConfig := cs_NR_HarqProcessConfig_None;
        v_ModulationScheme := qpsk;                                                                                     /* 38.214 clause 5.1.3.1:
                                                                                                                           The UE is not expected to decode a PDSCH scheduled with P-RNTI, RA-RNTI, SI-RNTI and Qm > 2
                                                                                                                           (Qm=2 corresponds to QPSK acc. 38.211 Table 7.3.1.2-1) */
      }
      case else {
        FatalError(__FILE__, __LINE__, "unknown or invalid DCI format");
      }
    }
    return cs_NR_DciDlInfo_Init(v_ResourceAllocationType,
                                v_FreqDomainResourceAssignmentDL,
                                v_TimeDomainResourceAssignment,
                                cs_NR_TransportBlockSchedulingDL_Automatic(v_ModulationScheme, p_RedundancyVersionList),
                                v_HarqProcessConfig,
                                p_DciFormatAndParams);
  }



2.2 [bookmark: _Toc144288535]function f_NR_RedcapWithIncreasedSib1Size
	Function name
	f_NR_RedcapWithIncreasedSib1Size

	Reason for change
	1. A new method is required to check the 
test cases having increased Sib1 size due to initialUplinkBWP-RedCap and initialDownlinkBWP-RedCap IEs.

	Summary of change
	1. Added a new method named f_NR_RedcapWithIncreasedSib1Size().


	TTCN module
	Common\NR\NR_ConfigurationSteps.ttcn

	MCC160 Comment
	



New Method
	  function f_NR_RedcapWithIncreasedSib1Size() return boolean //WA#WI=1098062
  {
    var charstring v_TestcaseName := testcasename();

       select (v_TestcaseName) {
      case ("TC_7_1_1_1_17_NR5GC") { return true; }
    }
    return false;
  }



2.3 [bookmark: _Toc144288536]template cads_NR_BWP_CONFIG_UL_RedCap_REQ
	Function name
	template cads_NR_BWP_CONFIG_UL_RedCap_REQ

	Reason for change
	1. Uninitialized parameters lead to run time exception.

	Summary of change
	1. Initialized remaining unused parameters to omit.

	TTCN module
	NR_TC_Common\7_1_1\MAC_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	  template (value) NR_SYSTEM_CTRL_REQ cads_NR_BWP_CONFIG_UL_RedCap_REQ(NR_CellId_Type  p_NR_CellId,
                                                                       template(value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                                                       template(value) NR_UplinkBWP_Type p_NR_UplinkBWP
                                                                       ) modifies cas_NR_CellConfigAllOmit_REQ :=
  {
    Request := {
      Cell := {
        AddOrReconfigure := {
          PhysicalLayer := {
            Common := omit,
            Downlink := omit,
            Uplink := {
              Uplink := {
                Config:= {
                  FrequencyInfoUL := omit,
                  BWPs := {
                    ActiveBWP := omit,
                    BwpArray := omit,
                    InitialBWP_RedCap := p_NR_UplinkBWP
                  },
                  RACH_ConfigDedicated := omit,
                  SI_RequestConfig := omit
                }
                
              }
            }
          }
        }
      }
    }
  };
  



After Change:
	  template (value) NR_SYSTEM_CTRL_REQ cads_NR_BWP_CONFIG_UL_RedCap_REQ(NR_CellId_Type  p_NR_CellId,
                                                                       template(value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                                                       template(value) NR_UplinkBWP_Type p_NR_UplinkBWP
                                                                       ) modifies cas_NR_CellConfigAllOmit_REQ :=
  {
    Request := {
      Cell := {
        AddOrReconfigure := {
          PhysicalLayer := {
            Common := omit,
            Downlink := omit,
            Uplink := {
              Uplink := {
                Config:= {
                  FrequencyInfoUL := omit,
                  BWPs := {
                    ActiveBWP := omit,
                    BwpArray := omit,
                    InitialBWP_RedCap := p_NR_UplinkBWP
                  },
                  RACH_ConfigDedicated := omit,
                  SI_RequestConfig := omit
                }
                
              },
              SupplementaryUplink     := omit,     // WA#WA=1098062
              TimingAdvance           := omit,     // WA#WA=1098062
              PUSCH_ServingCellConfig := omit,     // WA#WA=1098062
              PUSCH_ServingCellConfigSUL := omit   // WA#WA=1098062 
            }
          }
        }
      }
    }
  };



2.4 [bookmark: _Toc144288537]function f_TC_7_1_1_1_17_NR5GC
	Function name
	f_TC_7_1_1_1_17_NR5GC

	Reason for change
	1. In order to avoid BSR timer related issues due to UE capability being too long it is proposed to update the BSR time after the preamble (see R5s210862).
2. Comments at Step 1 and Step 12 to set useFullResumeID is irrelevant as it is not an rrcResume scenario.
3. template cs_PUCCH_ConfigCommon_711117 parameters are not aligned as per the test prose requirement. See section 2.5 for details.

	Summary of change
	1. Moved f_NR_CellInfo_SetMAC_CellGroupConfigL2() after    f_NR5GC_Preamble().
2. Deleted useFullResumeID comments at Step 1 and Step 12.
3. Used two new templates at Step 1 and Step 12 instead of using cs_PUCCH_ConfigCommon_711117 to align with test prose requirement.

	TTCN module
	NR5GC\7_1_1\MAC_R16_R17_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  function f_TC_7_1_1_1_17_NR5GC() runs on NR5GC_PTC
  {
    /* Random access procedure / RedCap UE identification */
    var template (value) NR_UplinkBWP_Type v_NR_UplinkBWP ; //Initial BWP
    var template (value) RACH_ConfigCommon v_RACH_ConfigCommon;
    var template (value) RACH_ConfigCommon v_RACH_ConfigCommon_Back;
    var template(omit) BWP_UplinkCommon.pucch_ConfigCommon v_Pucch_ConfigCommon_Back := omit;
   
    f_NR5GC_Init(NR_1);                                                 // Init Cell parameters
    f_NR_CellInfo_SetMAC_CellGroupConfigL2(nr_Cell1);
    f_NR_CellInfo_SetSIB1_PUCCHConfigCommon(nr_Cell1,cs_Pucch_ConfigCommon_Setup(cs_PUCCH_ConfigCommon_711117));
    f_NR_CellInfo_SetSIB1_FeaturePriority(nr_Cell1, cs_FeaturePriorities_711117);
  
    
    f_NR_CellConfig_Def(nr_Cell1);                                      // Create and configure NR cell 1
    f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A);

    f_NR_TestBody_Set(true);
    v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP (nr_Cell1, tsc_NR_BWP_Id); //Initial BWP
    v_Pucch_ConfigCommon_Back:= f_NR_CellInfo_GetSIB1_PUCCHConfigCommon( nr_Cell1);
    v_RACH_ConfigCommon := v_NR_UplinkBWP.Common.R15.rach_ConfigCommon.setup ;
    v_RACH_ConfigCommon_Back:= v_RACH_ConfigCommon;
   
    //@siclog "Step 1" siclog@
    //The SS changes the SIB1 of NR Cell 1 to set the useFullResumeID to True.
    //PUCCH_ConfigCommon & FeaturePriorities configured in preamble
   //@sic R5-231529 sic@
   v_RACH_ConfigCommon.featureCombinationPreamblesList_r17 := {cs_FeatureCombinationPreambles_711117(8)};
   f_NR_CellInfo_SetSIB1_InitialUplinkBWP_Rach_ConfigCommon(nr_Cell1,cs_Rach_ConfigCommon_Config_Setup(v_RACH_ConfigCommon) );
   
   //@siclog "Step 2-10" siclog@
   f_TC_7_1_1_1_17_NR_TestBody(8,3);
    //@siclog "Step 11" siclog@
    //The SS transmits RRCRelease
    f_NR_RRCRelease(nr_Cell1);
     //@siclog "Step  12" siclog@
    //The SS changes the SIB1 of NR Cell 1 to set the useFullResumeID to True.
   f_NR_CellInfo_SetSIB1_PUCCHConfigCommon(nr_Cell1, v_Pucch_ConfigCommon_Back);
   f_NR_CellInfo_SetSIB1_InitialUplinkBWP_Rach_ConfigCommon(nr_Cell1,cs_Rach_ConfigCommon_Config_Setup(v_RACH_ConfigCommon_Back) );
   v_RACH_ConfigCommon.featureCombinationPreamblesList_r17 := {cs_FeatureCombinationPreambles_711117(12)};
   f_NR_ConfigRedCapUL_BWP(nr_Cell1,cs_PUCCH_ConfigCommon_711117 ,v_RACH_ConfigCommon );
     //@siclog "Step 13-21" siclog@
    f_TC_7_1_1_1_17_NR_TestBody(12,3);
    
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  };



After Change:
	   function f_TC_7_1_1_1_17_NR5GC() runs on NR5GC_PTC
  {
    /* Random access procedure / RedCap UE identification */
    var template (value) NR_UplinkBWP_Type v_NR_UplinkBWP ; //Initial BWP
    var template (value) RACH_ConfigCommon v_RACH_ConfigCommon;
    var template (value) RACH_ConfigCommon v_RACH_ConfigCommon_Back;
    var template(omit) BWP_UplinkCommon.pucch_ConfigCommon v_Pucch_ConfigCommon_Back := omit;
   
    f_NR5GC_Init(NR_1);                                                 // Init Cell parameters
    
    f_NR_CellInfo_SetSIB1_PUCCHConfigCommon(nr_Cell1,cs_Pucch_ConfigCommon_Setup(cs_PUCCH_ConfigCommon_711117_Step1)); //WA#WA=1098062
    f_NR_CellInfo_SetSIB1_FeaturePriority(nr_Cell1, cs_FeaturePriorities_711117);
  
    
    f_NR_CellConfig_Def(nr_Cell1);                                      // Create and configure NR cell 1
    f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A);
    f_NR_CellInfo_SetMAC_CellGroupConfigL2(nr_Cell1); //WA#WA=1098062
    f_NR_TestBody_Set(true);
    v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP (nr_Cell1, tsc_NR_BWP_Id); //Initial BWP
    v_Pucch_ConfigCommon_Back:= f_NR_CellInfo_GetSIB1_PUCCHConfigCommon( nr_Cell1);
    v_RACH_ConfigCommon := v_NR_UplinkBWP.Common.R15.rach_ConfigCommon.setup ;
    v_RACH_ConfigCommon_Back:= v_RACH_ConfigCommon;
   
    //@siclog "Step 1" siclog@
     //PUCCH_ConfigCommon & FeaturePriorities configured in preamble
   //@sic R5-231529 sic@
   v_RACH_ConfigCommon.featureCombinationPreamblesList_r17 := {cs_FeatureCombinationPreambles_711117(8)};
   f_NR_CellInfo_SetSIB1_InitialUplinkBWP_Rach_ConfigCommon(nr_Cell1,cs_Rach_ConfigCommon_Config_Setup(v_RACH_ConfigCommon) );
   
   //@siclog "Step 2-10" siclog@
   f_TC_7_1_1_1_17_NR_TestBody(8,3);
    //@siclog "Step 11" siclog@
    //The SS transmits RRCRelease
    f_NR_RRCRelease(nr_Cell1);

     //@siclog "Step  12" siclog@
   f_NR_CellInfo_SetSIB1_PUCCHConfigCommon(nr_Cell1, v_Pucch_ConfigCommon_Back);
f_NR_CellInfo_SetSIB1_InitialUplinkBWP_Rach_ConfigCommon(nr_Cell1,cs_Rach_ConfigCommon_Config_Setup(v_RACH_ConfigCommon_Back) );
   v_RACH_ConfigCommon.featureCombinationPreamblesList_r17 := {cs_FeatureCombinationPreambles_711117(12)};
   f_NR_ConfigRedCapUL_BWP(nr_Cell1,cs_PUCCH_ConfigCommon_711117_Step12 ,v_RACH_ConfigCommon ); //WA#WA=1098062
     //@siclog "Step 13-21" siclog@
    f_TC_7_1_1_1_17_NR_TestBody(12,3);
    
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  };




2.5 [bookmark: _Toc144288538]templates cs_PUCCH_ConfigCommon_711117_Step1, cs_PUCCH_ConfigCommon_711117_Step12
	Function name
	Templates cs_PUCCH_ConfigCommon_711117_Step1, cs_PUCCH_ConfigCommon_711117_Step12

	Reason for change
	Current tempate cs_PUCCH_ConfigCommon_711117 is not suited with test prose requirement at STEP 1 for initialUplinkBWP and STE12 for initialUplinkBWP-RedCap as per the Section Table 7.1.1.1.17.3.3-6.

	Summary of change
	Created two new templates and used those at Step1 and Step12.

	TTCN module
	NR_TC_Common\7_1_1\MAC_R16_R17_TC_Common_NR.ttcn

	MCC160 Comment
	



New Templates: 
	
template(value) PUCCH_ConfigCommon cs_PUCCH_ConfigCommon_711117_Step1 :=
    { //Acc to Table 7.1.1.1.17.3.3-6
    pucch_ResourceCommon  := 0,
    pucch_GroupHopping    := enable,
    hoppingId             := omit,
    p0_nominal            := -90,
    nrofPRBs              := omit,
    intra_SlotFH_r17      := omit,
    pucch_ResourceCommonRedCap_r17 := omit, //WA#WA=1098062
    additionalPRBOffset_r17        := omit
    };


template(value) PUCCH_ConfigCommon cs_PUCCH_ConfigCommon_711117_Step12 :=
    { //Acc to Table 7.1.1.1.17.3.3-6
    pucch_ResourceCommon  := omit,
    pucch_GroupHopping    := enable,
    hoppingId             := omit,
    p0_nominal            := -90,
    nrofPRBs              := omit,
    intra_SlotFH_r17      := omit, //fromLowerEdge,
    pucch_ResourceCommonRedCap_r17 := 0, //WA#WA=1098062
    additionalPRBOffset_r17        := omit //n2
    };



2.6 [bookmark: _Toc144288539]function f_TC_7_1_1_1_17_NR_TestBody
	Function name
	f_TC_7_1_1_1_17_NR_TestBody

	Reason for change
	1. It is not an rrcResume scenario thus UE is not in RRC_INACTIVE state.
2. ULGrant configurations are missing at Step 8, 19.

	Summary of change
	1. Corrected the state at Step 2, 13, and Step 4, 15 steps code.
2. Added f_NR_ULGrantConfiguration_Start() at Step 8, 19.

	TTCN module
	NR5GC\Common\MAC_Common_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	 function f_TC_7_1_1_1_17_NR_TestBody(integer p_StartPreamble,
                                       integer p_NumberOfPreamble) runs on NR5GC_PTC
  {
    var integer v_ModificationPeriodFrames;
    var float v_ModificationPeriod;
    var template (omit) NR_SRB_COMMON_IND v_SrbIndDummy := omit;
    var template (omit) NR_SYSTEM_IND v_SysIndDummy1 := omit;
    var template (omit) NR_SYSTEM_IND v_SysIndDummy2 := omit;
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    //@siclog "Step 2, 13" siclog@
    //The SS transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.
    f_NR_ModifySysinfo(nr_Cell1, true, RRC_INACTIVE);

    //@siclog "Step 3, 14" siclog@
    //Wait for 2.1* modification period second for the UE to receive new system information.
    v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod(nr_Cell1);
    v_ModificationPeriod := int2float(v_ModificationPeriodFrames)/100.0; //Modification period in seconds
    f_Delay(2.1 * v_ModificationPeriod);

    //Configure SS to report UL_SRB0_LCID
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_SRB0_LCID(enable));
    // Enable PRACH reception indication
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));

    //@siclog "Step 4, 15 " siclog@
    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).
    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    //@siclog "Step 5-7, 16-18" siclog@
    //Check: Does the UE transmit Preamble on PRACH with ra-PreambleIndex range from 8 to 11
   // Check: Does the UE transmit a MAC PDU containing an RRCSetupRequest message by setting LCID to 35
    a_NR_RacingCond_AwaitRrcMessageAndTwoSysInd(nr_Cell1,
                                                SysInd1BeforeMsgBeforeSysInd2,
                                                v_SrbIndDummy,
                                                v_SysIndDummy1,
                                                v_SysIndDummy2,
                                                car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest),
                                                car_NR_PRACH_Preamble_IND(nr_Cell1, -, (p_StartPreamble..p_StartPreamble+p_NumberOfPreamble)),
                                                cr_NR_SystemIndicationUL_SRB0_LCID_IND(nr_Cell1, cr_TimingInfo_Any, 35)
                                             );
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5, 7, 16, 18");
    //Configure SS to stop report UL_SRB0_LCID
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_SRB0_LCID(disable));
     // Configure SS to stop PRACH reception indication
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
    
    //@siclog "Step 8, 19" siclog@
     //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 9, 20" siclog@
     //The UE transmit an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);
    //@siclog "Step 10, 21" siclog@
    //Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
  
  };

}   



After Change:
	
  function f_TC_7_1_1_1_17_NR_TestBody(integer p_StartPreamble,
                                       integer p_NumberOfPreamble) runs on NR5GC_PTC
  {
    var integer v_ModificationPeriodFrames;
    var float v_ModificationPeriod;
    var template (omit) NR_SRB_COMMON_IND v_SrbIndDummy := omit;
    var template (omit) NR_SYSTEM_IND v_SysIndDummy1 := omit;
    var template (omit) NR_SYSTEM_IND v_SysIndDummy2 := omit;
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    //@siclog "Step 2, 13" siclog@
    //The SS transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.
    f_NR_ModifySysinfo(nr_Cell1, true); // WA#WA=1098062

    //@siclog "Step 3, 14" siclog@
    //Wait for 2.1* modification period second for the UE to receive new system information.
    v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod(nr_Cell1);
    v_ModificationPeriod := int2float(v_ModificationPeriodFrames)/100.0; //Modification period in seconds
    f_Delay(2.1 * v_ModificationPeriod);

    //Configure SS to report UL_SRB0_LCID
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_SRB0_LCID(enable));
    // Enable PRACH reception indication
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));

    //@siclog "Step 4, 15 " siclog@
    //The SS transmits a Paging message including a matched identity.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE); // WA#WA=1098062

    //@siclog "Step 5-7, 16-18" siclog@
    //Check: Does the UE transmit Preamble on PRACH with ra-PreambleIndex range from 8 to 11
   // Check: Does the UE transmit a MAC PDU containing an RRCSetupRequest message by setting LCID to 35
    a_NR_RacingCond_AwaitRrcMessageAndTwoSysInd(nr_Cell1,
                                                SysInd1BeforeMsgBeforeSysInd2,
                                                v_SrbIndDummy,
                                                v_SysIndDummy1,
                                                v_SysIndDummy2,
                                                car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest),
                                                car_NR_PRACH_Preamble_IND(nr_Cell1, -, (p_StartPreamble..p_StartPreamble+p_NumberOfPreamble)),
                                                cr_NR_SystemIndicationUL_SRB0_LCID_IND(nr_Cell1, cr_TimingInfo_Any, 35)
                                             );
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5, 7, 16, 18");
    //Configure SS to stop report UL_SRB0_LCID
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_SRB0_LCID(disable));
     // Configure SS to stop PRACH reception indication
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
    
    //@siclog "Step 8, 19" siclog@
    f_NR_ULGrantConfiguration_Start(nr_Cell1, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // WA#WA=1098062
     //The SS transmits an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell1);

    //@siclog "Step 9, 20" siclog@
     //The UE transmit an RRCSetupComplete message including SERVICE REQUEST to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);
    //@siclog "Step 10, 21" siclog@
    //Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are executed to successfully complete the service request procedure.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
  
  };

}



1. [bookmark: _Toc54888065][bookmark: _Toc144288540][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]Branches executed
This NR5GC TC was executed in Band combination “n41” using EEA1/EIA1 algorithms”.
1. [bookmark: _Toc144288541]Execution Log Files 
4. [bookmark: _Toc144288542]MediaTek
The MediaTek MT6815 UE passed this test case on R&S® 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
Test case execution log file:
tc_7_1_1_1_17_MTK.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
1. [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc144288543]References
	[1]
	R5s230657:   Supporting information for addition of NR5GC Redcap testcase 7.1.1.1.17 in FR1





